AD-A146 181 DESCRIPTION AND USE OF THE SINGLE-COLOR TRANSMISSOMETER 1/1
PLUME DIAGNOSTIC..(U) AEROSPACE CORP EL SEGUNDO CA
S J YOUNG AUG 84 TR-0084(4623-02)-1 AFRPL-TR-84-048
UNCLASSIFIED F04701-83-C-0084 F/G 21/2 NL

END
FILMED
DTIC



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS-1963-A

(12)

AFRPL TR-84-048 AD:

Final Report
for the period
April 1983 to
June 1984

Description and use of the Single-Color Transmissometer Plume Diagnostic Code A32CODE

August 1984

Authors:
S. J. Young

The Aerospace Corporation
El Segundo CA 90245

TR-0084(4623-02)-1
F04701-83-C-0084

AD-A146 181

Approved for Public Release

prepared for the: **Air Force Rocket Propulsion Laboratory**

Air Force Space Technology Center
Space Division, Air Force Systems Command
Edwards Air Force Base,
California 93523

NOTICE

When U.S. Government drawings, specifications, or other data are used for any purpose other than a definitely related Government procurement operation, the Government thereby incurs no responsibility nor any obligation whatsoever, and the fact that the Government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data is not to be regarded by implication or otherwise, or in any manner licensing the holder or any other person or corporation, or conveying any rights or permission to manufacture, use, or sell any patented invention that may be related thereto.

FOREWORD

This manual was submitted by Aerospace Corporation, El Segundo, California 90245, under Contract No. F04701-83-C-0084 with the Air Force Rocket Propulsion Laboratory, Edwards AFB, California 93523, under Air Force Project Task 314800AP. The manual gives detailed instructions for running the Single-Color Transmissometer Plume Diagnostic Code (A32CODE). The technical development of the User's Manual has also been published as AFRPL-TR-84-048.

This report has been reviewed and approved for publication in accordance with the distribution statement on the cover and on the DD form 1473.

James A. Misener

JAMES A. MISENER, 2LT, USAF
Project Manager

L. Kevin Slimak

L. KEVIN SLIMAK
Chief, Interdisciplinary Space Tech Br

FOR THE DIRECTOR

Robert L. Geisler

ROBERT L. GEISLER
Chief, Propulsion Analysis Division

ADA146181

SECURITY CLASSIFICATION OF THIS PAGE

REPORT DOCUMENTATION PAGE

1a. REPORT SECURITY CLASSIFICATION UNCLASSIFIED	1b. RESTRICTIVE MARKINGS
2a. SECURITY CLASSIFICATION AUTHORITY	3. DISTRIBUTION/AVAILABILITY OF REPORT Approved for Public Release; Distribution Unlimited
2b. DECLASSIFICATION/DOWNGRADING SCHEDULE	
4. PERFORMING ORGANIZATION REPORT NUMBER(S) TR-0084(4623-02)-1	5. MONITORING ORGANIZATION REPORT NUMBER(S) AFRPL-TR-84-048
6a. NAME OF PERFORMING ORGANIZATION The Aerospace Corporation 6b. OFFICE SYMBOL (If applicable)	7a. NAME OF MONITORING ORGANIZATION Air Force Rocket Propulsion Laboratory
6c. ADDRESS (City, State and ZIP Code) El Segundo CA 90245	7b. ADDRESS (City, State and ZIP Code) Stop 24 Edwards AFB CA 93523
8a. NAME OF FUNDING/SPONSORING ORGANIZATION Space Division 8b. OFFICE SYMBOL (If applicable)	9. PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER F04701-83-C-0084
8c. ADDRESS (City, State and ZIP Code) Los Angeles Air Force Station Los Angeles CA 90009	10. SOURCE OF FUNDING NOS. PROGRAM ELEMENT NO.: 62302F PROJECT NO.: 3148 TASK NO.: 00 WORK UNIT NO.: AP
11. TITLE (Include Security Classification) Description and Use of the Single-Color Transmissometer Plume ...	

12. PERSONAL AUTHOR(S)
Young, Stephen J.

13a. TYPE OF REPORT	13b. TIME COVERED	14. DATE OF REPORT (Yr., Mo., Day)	15. PAGE COUNT
Final	FROM 83/04 TO 84/06	84/08	36

16. SUPPLEMENTARY NOTATION

17. COSATI CODES			18. SUBJECT TERMS (Continue on reverse if necessary and identify by block number)
FIELD	GROUP	SUB. GR.	Plume diagnostics
21	08		Plume particulates
			Retrieval diagnostics

19. ABSTRACT (Continue on reverse if necessary and identify by block number)

A description of and instructions for use of the code A32CODE for analysis of AFRPL single-color transmissometer data are presented.

20. DISTRIBUTION/AVAILABILITY OF ABSTRACT UNCLASSIFIED/UNLIMITED ☒ SAME AS RPT. ☐ DTIC USERS ☐	21. ABSTRACT SECURITY CLASSIFICATION Unclassified
22a. NAME OF RESPONSIBLE INDIVIDUAL Lt James A. Misener	22b. TELEPHONE NUMBER (Include Area Code) (805) 277-5584 22c. OFFICE SYMBOL AFRPL/DYSO

DD FORM 1473, 83 APR   EDITION OF 1 JAN 73 IS OBSOLETE.

SECURITY CLASSIFICATION OF THIS PAGE

Block 11: Diagnostic Code A32CODE (U).

CONTENTS

FIGURES

TABLES

1. INTRODUCTION

The computer code A32CODE performs automatic analysis of the AFRPL single-color transmissometer data to obtain the volume-to-surface mean radius a_{32} of plume particles and estimates of its error caused by uncertainty in size distribution, complex index of refraction, and experimental data. The code consists of two parts: the actual program (A32CODE) and subprograms for analyzing the data, and files of Mie scattering parameters (UQFILE and BQFILE).

The method of analysis has been presented in detail in Section 4.1 of Ref. 1 and is only briefly reviewed here in the discussions of how the code works. The input data required by the code are defined in Table 1. With the data of the first card and the data of either UQFILE (if DISTR = U) or BQFILE (if DISTR = B), the upper and lower error bounds of $\bar{Q}/a_{32}$ versus a_{32} are constructed and stored. With the data from the second and subsequent cards, the experimentally determined value of $\bar{Q}/a_{32}$ is computed and used with the constructed curves to determine a_{32} and its error. An example application is described in a later section.

---

1. S. J. Young, Considerations on the Retrieval of Plume Particle Properties from the AFRPL Transmissometer and Polarization Scattering Experiments, AFRPL-TR- 84 - 047 , U. S. Air Force Rocket Propulsion Laboratory, Edwards Air Force Base, California, August 1984.

Table 1. A32CODE Card Input Format

Columns 1–10	11–20	21–30	31–40	41–50	51–60	61–70	71–80
n_1	n_2	κ_1	κ_2	λ	DISTR		
τ	$\Delta\tau$	L	ΔL	C_m	ΔC_m	d	Δd
τ	$\Delta\tau$	...	repeat experimental data cards as desired				
⋮							
Terminate with EOI card							

All fields are E10 except DISTR which is A10.

n_1 — lower bound of real part of refractive index
n_2 — upper bound of real part of refractive index
κ_1 — lower bound of imaginary part of refractive index
κ_2 — upper bound of imaginary part of refractive index

$1.60 \leq n_1 \leq n_2 \leq 2.00$
$0 \leq \kappa_1 \leq \kappa_2 \leq 0.50$

λ — wavelength (μm)
DISTR = U (right justified) implies analysis with unimodal rectangular size distribution
DISTR = B (right justified) implies analysis with bimodal rectangular size distribution

τ, $\Delta\tau$ — measured transmittance and error
L, ΔL — plume diameter and error (cm)
C_m, ΔC_m — mass loading and error (g/cm^3)
d, Δd — particulate material bulk density and error (g/cm^3)

2. MIE PARAMETER FILES

The two files UQFILE and BQFILE contain the Mie scattering parameter $\bar{Q}/x_{32}$ averaged over all possible unimodal and bimodal rectangular size distributions, respectively, for which x_{32} is fixed at the tabulated value. The files contain entries for the 54 pairs of n, κ values implied by

$$n = 1.60,\ 1.65,\ 1.70,\ 1.75,\ 1.80,\ 1.85,\ 1.90,\ 1.95,\ 2.00$$
$$\kappa = 0,\ 2 \times 10^{-4},\ 2 \times 10^{-3},\ 2 \times 10^{-2},\ 0.2,\ 0.5$$

The structure of either file is defined by the FORTRAN read sequence of subroutine READQ (a listing of the code is given in the Appendix). NINDEX(N) (N=1,9) and KINDEX(K) (K=1,6) are the arrays just listed. X32(I) (I=1,41) is the array

$$x_{32}(i) = 10^{0.1} x_{32}(i-1);\ i=1,41;\ x_{32}(1) = 0.1$$

QBAR(N,K,I) is the value of Q/x_{32} averaged over the size distribution, QSIG(N,K,I) is the standard deviation of variation about the mean value, QMIN(N,K,I) is the minimum value in the variation, and QMAX(N,K,I) is the maximum value in the variation.

3. A32CODE DESCRIPTION

The first step in A32CODE is to read the input data of the first data card and the data of either UQFILE or BQFILE. The latter is done by subroutine READQ. The data of the first card specify the degree of ignorance assumed in n and κ, the wavelength, and whether the analysis is to be done using a unimodal or bimodal rectangular particle size distribution (see Table 1 for details).

The next step is to effect a variation over n and κ. The ranges of n and κ are divided into NN-1 and KK-1 equal size subintervals and the five Q-variables described above are computed at the NN× KK grid points for all 41 values of x_{32} (the code uses NN=KK=5, but these numbers can easily be changed). The computation of the Q values consists of interpolations on the values in either the UQFILE or BQFILE. The interpolation is of the form

$$\text{Log } Q = (A+Bn) + (C+Dn)\kappa \tag{1}$$

and is performed in subroutine QINT. At each of the 41 values of x_{32}, two upper and two lower bounds of $\bar{Q}/a_{32}$ are determined. These are

$$\begin{aligned} \text{QMAX1} &= \max\,(\text{QBAR+QSIG})\ c \\ \text{QMAX2} &= \max\,(\text{QMAX})\ c \\ \text{QMIN1} &= \min\,(\text{QBAR-QSIG})\ c \\ \text{QMIN2} &= \min\,(\text{QMIN})\ c \end{aligned} \tag{2}$$

where max ( ) and min ( ) denote the maximum and minimum values, respectively, in the NN× KK set of variation values. QMAX1 and QMIN1 represent "optimistic" error bounds in that they are the traditional root-mean-square (rms) deviations from the mean. QMAX2 and QMIN2 represent "pessimistic" error bounds in that they are the absolute maximum and minimum values in the variation over index of refraction and size distribution. If QMIN1 is less than or equal to zero, it is replaced with QMIN2. The constant c is $2\pi/\lambda$ and transforms $\bar{Q}/x_{32}$ into $\bar{Q}/a_{32}$ with unit μm^{-1}.

The end result of the variation then, is two sets of upper and lower bound curves versus a_{32}. In the code, these curves are denoted by

$$\left.\begin{array}{l} BL(m,i) \\ BU(m,i) \end{array}\right\} i=1,41$$

where L implies "lower", U implies "upper", m=1 implies rms error and m=2 implies full-range error. These curves are essentially the bound curves of, for example, Fig. 31 of Ref. 1. The four curves are plotted versus a_{32} in subroutine PLOTB (note, $x_{32} = 2\pi a_{32}/\lambda$).

For subsequent analysis, it is convenient to have an algebraic representation of these curves. Here, a quadratic fit is made on the logarithm of the curves. In each interval a_i to a_{i+1} with i=1,40, the curves are approximated by [B(x) is any one of the bound curves and a is a_{32}]

$$B_i(a) = e^{\gamma_i + \beta_i \ln (a/a_i) + \alpha_i \ln^2(a/a_i)} \tag{3}$$

where

$$\alpha_i = \frac{1}{2\ln^2(a_{i+1}/a_i)} \ln \left(\frac{B_i \; B_{i+2}}{B_{i+1}^2}\right) \tag{4a}$$

$$\beta_i = \frac{1}{2\ln (a_{i+1}/a_i)} \ln \left(\frac{B_{i+1}^4}{B_i^3 \; B_{i+2}}\right) \tag{4b}$$

$$\gamma_i = \ln B_i \tag{4c}$$

for i=1,39, and

$$\alpha_i = \alpha_{39} \qquad \beta_i = \beta_{39} \qquad \gamma_i = \gamma_{39} \tag{5}$$

for i=40.

In the code, these coefficient arrays are denoted by

$$\left.\begin{array}{l} ACL(m,i) \\ BCL(m,i) \\ CCL(m,i) \\ ACU(m,i) \\ BCU(m,i) \\ CCU(m,i) \end{array}\right\} \begin{array}{l} i=1,40 \\ m=1,2 \text{ (i.e., rms or full-range error)} \end{array}$$

This curve fitting is done in subroutine CURFIT. In addition to the curve fitting, the peak value of each curve is computed in CURFIT and stored as

$$\left.\begin{array}{l} \text{YLMAX(m)} \\ \\ \text{YUMAX(m)} \end{array}\right\} \quad m=1,2$$

At this point in the code, a loop is initiated over experimental input. One card at a time is read in, the data contained on it are analyzed, and the results for mean particle size and error printed out. The loop continues until an end-of-information card is reached, at which point, the code stops.

The experimental input data required are described in Table 1. From these data, the experimental value of $\bar{Q}/a_{32}$ (denoted by Y in the code) is computed from [see Eq. (42) in Ref. 1]

$$Y(\text{in } \mu m^{-1}) = -\frac{4d}{3LC_m} 10^{-4} \ln \tau \tag{6}$$

Its experimental error is determined from

$$DY = Y\sqrt{\left(\frac{\Delta L}{L}\right)^2 + \left(\frac{\Delta C_m}{C_m}\right)^2 + \left(\frac{\Delta d}{d}\right)^2 + \left(\frac{\Delta \tau}{\tau \ln \tau}\right)^2} \tag{7}$$

The experimental bounds on $\bar{Q}/a_{32}$ are then defined by

$$\begin{aligned} YP &= Y + DY \\ YM &= Y - DY \end{aligned} \tag{8}$$

At this point, an analysis is made of where these two bounds (ie., YP and YM) intersect the BU and BL curves. (A loop is made over m=1 and 2 so that the following is done for both the rms and full-range error bound cases). The primary differentiation on the manner of analysis is made by testing the magnitudes of YP and YM with YUMAX and YLMAX. The most general case occurs when YP is less than YLMAX, that is, when the upper experimental bound lies below the peak of the lower bound curve. This condition is shown in Fig. 1. In this case there are two well-defined regions of the a_{32} axis where the experimental data are consistent with the analysis curves. They lie on either side of the peaks of the curves BU and BL and are designated here as lying in



Fig. 1. Intersection Diagram for Retrieval of a_{32} and Error

the "upper" or "lower" branch of the analysis curves. The eight points of intersection of YP and YM with BU and BL shown in Fig. 1 are computed in subroutine ROOTS (described later). The retrieved a_{32} and errors are defined in terms of the a_{32} coordinates of these points by

$$a_{32} = \begin{cases} \frac{1}{4}\,[a^{+}_{U1} + a^{+}_{L1} + a^{-}_{U1} + a^{-}_{L1}] & \text{on lower branch} \\ \frac{1}{4}\,[a^{+}_{L2} + a^{+}_{U2} + a^{-}_{L2} + a^{-}_{U2}] & \text{on upper branch} \end{cases} \tag{9a}$$

$$\Delta a_{32} = \begin{cases} \frac{1}{2}\,(a^{+}_{L1} - a^{-}_{U1}) & \text{on lower branch} \\ \frac{1}{2}\,(a^{-}_{U2} - a^{+}_{L2}) & \text{on upper branch} \end{cases} \tag{9b}$$

(The code symbols for these intersection points are AU1P, AL1P, AU1M, AL1M, AL2P, AU2P, AL2M and AU2M).

In most cases, it will be known from experimental conditions which of the two retrieved values is relevant. For those cases where it is not known, the code also computes a "branch undetermined" retrieval value for a_{32}. In principle, this value and its error could be determined from the four intersections of YP and YM and BU by

$$a_{32} = \frac{1}{2}\left[\frac{a^{+}_{U1} + a^{-}_{U1}}{2} + \frac{a^{+}_{U2} + a^{-}_{U2}}{2}\right] \tag{10}$$

$$\Delta a_{32} = \frac{1}{2}\,(a^{-}_{U2} - a^{-}_{U1})$$

In practice, however, these values are more conveniently computed in the code from the intersections of the mean experimental value Y with the upper bound curve BU (Fig. 1) by

$$a_{32} = \frac{1}{2}\,(a_1 + a_2)$$

$$\Delta a_{32} = \frac{1}{2}\,(a_2 - a_1) \tag{11}$$

(The code symbols for these intersection points are A1 and A2.)

If the upper experimental bound YP is greater than the peak YLMAX of the lower bound curve, there are no well-defined "upper" and "lower" branch regions. For this case, a "branch undefined" analysis is made exactly as for a "branch undetermined" case. That is, the intersections of Y with BU are determined and Eqs. (11) used to define the retrieval a_{32} and error. A minor variation is made if Y is greater than YUMAX (so that there are no intersections of Y with BU) — the value of YM is used in place of Y. If YM is greater the YUMAX, no analysis is possible. (Also, analysis is skipped if YM is less than the lowest value of BL in the entire a_{32} range.)

For all cases of analysis, the intersection of a line Y (i.e., Y, YP, or YM) with a bound curve B (i.e., BU or BL) is made in subroutine ROOTS using the algebraic representation for B defined by Eqs. (3)-(5). In each of the 40 subintervals, a test is made to see if there are any intersections of Y with the quadratic form of Eq. (3). There will be no real intersections if

$$F = \beta_i^2 - 4\alpha_i\,(\gamma_i - \ln Y) < 0$$

and two intersections (they may be equal) otherwise. These intersections are

$$a^{(1,2)} = a_i\, e^{(\sqrt{F} \pm \beta_i)/2\alpha_i}$$

A further check is made to see if these roots occur in the interval a_i to a_{i+1}. If they do, they are saved. After all roots in all 40 subintervals have been found (there may be many roots if B fluctuates up and down a lot) a search is made for the smallest and largest. These two are returned to the main code as the relevant intersection points.

4. EXAMPLE APPLICATION

Example runs of the code were made for the case where ignorance of the particulate index of refraction is specified by

$$1.7 < n < 1.8$$
$$0 < \kappa < 0.02$$

and λ is the AFRPL experimental value $\lambda = 0.5145$ μm.

Both the unimodal and bimodal size distribution were used. Two cases of experimental data were run. In both,

$$\tau = 0.880$$
$$L = 10 \text{ cm}$$
$$C_m = 5.5 \times 10^{-7} \text{g/cm}^3$$
$$d = 3.7 \text{ g/cm}^3$$

In the first case, these values were assumed to be exact and the input errors were set to zero. The resulting error in a_{32} is then due soley to ignorance of size distribution and index of refraction. In the second case, the following experimental errors were assigned:

$\Delta\tau = 0.03$ (~2% error in I and I_o; $\tau = I/I_o$)

$\Delta L = 1$ cm (10%)

$\Delta C_m = 5.5 \times 10^{-8}$ g/cm^3 (10%)

$\Delta d = 0.4$ g/cm^3 (~10%)

Code results for these two cases with the unimodal size distribution are shown in Tables 2 and 3 and Figs. 2 and 3. Table 2 lists the $\bar{Q}/a_{32}$ error bounds. AVE±SIG are the "optimistic" error bounds wherein the variation over size distribution is measured by the standard deviation of $\bar{Q}/a_{32}$; MAX and MIN are the "pessimistic" error bounds wherein the variation over size distribution is measured by the total range of variation of $\bar{Q}/a_{32}$. The choice of whether to be optimistic or pessimistic is a user option. These bound results are plotted in Figs. 2 and 3. (Note: the Aerospace plotting routines

Table 2. $\bar{Q}/a_{32}$ Bounds Listing for Unimodal Distribution

A32CODE RESULTS

```
NLIMITS          N1= 1.700E+00    N2= 1.800E+00
KLIMITS          K1= 0.           K2= 2.000E-02

WAVELENGTH=   .5145 MICRON
DISTRIBUTION=           U
```

Q/A32 BOUNDS (UNIT=1/MU)

I	A32(MU)	X32	AVE+SIG	AVE-SIG	MAX	MIN
1	8.189E-03	1.000E-01	4.271E-01	5.241E-03	4.273E-01	4.875E-03
2	1.032E-02	1.260E-01	4.359E-01	1.050E-02	4.361E-01	9.770E-03
3	1.294E-02	1.580E-01	4.521E-01	2.074E-02	4.526E-01	1.930E-02
4	1.638E-02	2.000E-01	4.846E-01	4.222E-02	4.854E-01	3.925E-02
5	2.055E-02	2.510E-01	5.454E-01	8.389E-02	5.469E-01	7.794E-02
6	2.588E-02	3.160E-01	6.847E-01	1.687E-01	6.883E-01	1.567E-01
7	3.259E-02	3.980E-01	9.777E-01	3.411E-01	9.849E-01	3.163E-01
8	4.102E-02	5.010E-01	1.567E+00	6.908E-01	1.581E+00	6.409E-01
9	5.167E-02	6.310E-01	2.747E+00	1.402E+00	2.776E+00	1.302E+00
10	6.502E-02	7.940E-01	4.955E+00	2.795E+00	5.005E+00	2.616E+00
11	8.189E-02	1.000E+00	8.621E+00	5.262E+00	8.712E+00	5.029E+00
12	1.032E-01	1.260E+00	1.546E+01	9.099E+00	1.553E+01	8.649E+00
13	1.294E-01	1.580E+00	2.098E+01	1.547E+01	2.231E+01	1.545E+01
14	1.638E-01	2.000E+00	2.258E+01	1.825E+01	2.304E+01	1.809E+01
15	2.055E-01	2.510E+00	2.157E+01	1.768E+01	2.282E+01	1.768E+01
16	2.588E-01	3.160E+00	1.709E+01	1.311E+01	1.889E+01	1.296E+01
17	3.259E-01	3.980E+00	1.067E+01	8.138E+00	1.140E+01	7.843E+00
18	4.102E-01	5.010E+00	6.724E+00	5.015E+00	6.645E+00	4.148E+00
19	5.167E-01	6.310E+00	5.494E+00	4.641E+00	5.679E+00	4.312E+00
20	6.502E-01	7.940E+00	4.482E+00	3.681E+00	5.154E+00	3.681E+00
21	8.189E-01	1.000E+01	3.167E+00	2.636E+00	3.081E+00	2.225E+00
22	1.032E+00	1.260E+01	2.447E+00	2.208E+00	2.628E+00	2.121E+00
23	1.294E+00	1.580E+01	1.862E+00	1.768E+00	1.930E+00	1.744E+00
24	1.638E+00	2.000E+01	1.415E+00	1.354E+00	1.450E+00	1.305E+00
25	2.055E+00	2.510E+01	1.132E+00	1.071E+00	1.206E+00	1.050E+00
26	2.588E+00	3.160E+01	8.564E-01	8.357E-01	8.602E-01	8.147E-01
27	3.259E+00	3.980E+01	6.748E-01	6.573E-01	6.879E-01	6.443E-01
28	4.102E+00	5.010E+01	5.312E-01	5.154E-01	5.433E-01	5.019E-01
29	5.167E+00	6.310E+01	4.136E-01	4.076E-01	4.164E-01	4.015E-01
30	6.502E+00	7.940E+01	3.286E-01	3.230E-01	3.351E-01	3.209E-01
31	8.189E+00	1.000E+02	2.576E-01	2.538E-01	2.606E-01	2.510E-01
32	1.032E+01	1.260E+02	2.026E-01	2.004E-01	2.031E-01	1.984E-01
33	1.294E+01	1.580E+02	1.608E-01	1.591E-01	1.621E-01	1.577E-01
34	1.638E+01	2.000E+02	1.265E-01	1.255E-01	1.273E-01	1.255E-01
35	2.055E+01	2.510E+02	9.999E-02	9.958E-02	1.002E-01	9.930E-02
36	2.588E+01	3.160E+02	7.907E-02	7.883E-02	7.915E-02	7.865E-02
37	3.259E+01	3.980E+02	6.269E-02	6.242E-02	6.281E-02	6.226E-02
38	4.102E+01	5.010E+02	4.959E-02	4.951E-02	4.968E-02	4.951E-02
39	5.167E+01	6.310E+02	3.929E-02	3.922E-02	3.932E-02	3.921E-02
40	6.502E+01	7.940E+02	3.116E-02	3.110E-02	3.114E-02	3.106E-02
41	8.189E+01	1.000E+03	2.470E-02	2.467E-02	2.473E-02	2.467E-02

Table 3. a_{32} Retrieval Results for Unimodal Distribution

A32 AND TOTAL ERROR RESULTS

FIRST ROW=VALUE
SECOND ROW=RMS ERROR
THIRD ROW=PERCENT ERROR

LB=LOWER BRANCH
UB=UPPER BRANCH
0B=UNDETERMINED BRANCH

ZERO=NOT APPLICABLE

TRANS	L(CM)	C(G/CM3)	D(G/CM3)	Q/A(MU-1)	A32(MU) RMS ERROR BOUNDS LB	UB	0B	A32(MU) FULL ERROR BOUNDS LB	UB	0B
8.800E-01	1.000E+01	5.500E-07	3.700E+00	1.147E+01	1.01E-01	2.95E-01	2.02E-01	1.02E-01	3.00E-01	2.08E-01
0.	0.	0.	0.	0.	1.08E-02	1.92E-02	1.12E-01	1.14E-02	2.48E-02	1.17E-01
0.	0.	0.	0.	0.	1.07E+01	6.49E+00	5.53E+01	1.12E+01	8.25E+00	5.65E+01
8.800E-01	1.000E+01	5.500E-07	3.700E+00	1.147E+01	1.01E-01	3.04E-01	2.02E-01	1.02E-01	3.06E-01	2.08E-01
3.000E-02	1.000E+00	5.500E-08	4.000E-01	3.676E+00	2.48E-02	7.00E-02	1.12E-01	2.51E-02	7.08E-02	1.17E-01
2.667E+01	1.000E+01	1.000E+01	1.081E+01	3.206E+01	2.45E+01	2.30E+01	5.53E+01	2.46E+01	2.31E+01	5.65E+01



Fig. 2. $\bar{Q}/a_{32}$ Bounds - Unimodal Distribution, rms Error



Fig. 3. $\bar{Q}/a_{32}$ Bounds - Unimodal Distribution, Full-Range Error

are inhouse and not transferable to other computer facilities; the plot routine in the delivered version of the code gets everything ready to plot, but does not actually plot them.)

Table 3 lists the actual retrieved a_{32} and its error for both the optimistic and pessimistic error bounds. For each of these, three values are determined depending on which branch (if either) of the $\bar{Q}/a_{32}$ curve is chosen. For the AFRPL conditions, it is probably safe to assume the upper branch.

Tables 4 and 5 and Figs. 4 and 5 repeat the results for the bimodal size distribution.

Table 4. $\bar{Q}/a_{32}$ Bounds Listing for Bimodal Distribution

A32CODE RESULTS

NLIMITS	N1= 1.700E+00	N2= 1.800E+00
KLIMITS	K1= 0.	K2= 2.000E-02

WAVELENGTH= .5145 MICRON
DISTRIBUTION= B

Q/A32 BOUNDS (UNIT=1/MU)

I	A32(MU)	X32	AVE+SIG	AVE-SIG	MAX	MIN
1	8.189E-03	1.000E-01	4.484E-01	4.875E-03	5.659E-01	4.875E-03
2	1.032E-02	1.260E-01	4.754E-01	9.771E-03	7.347E-01	9.771E-03
3	1.294E-02	1.580E-01	5.285E-01	1.910E-02	1.074E+00	1.910E-02
4	1.638E-02	2.000E-01	6.535E-01	3.925E-02	1.743E+00	3.925E-02
5	2.055E-02	2.510E-01	8.936E-01	7.732E-02	2.870E+00	7.732E-02
6	2.588E-02	3.160E-01	1.331E+00	1.567E-01	4.901E+00	1.567E-01
7	3.259E-02	3.980E-01	2.020E+00	3.164E-01	7.973E+00	3.164E-01
8	4.102E-02	5.010E-01	2.960E+00	6.408E-01	1.375E+01	6.408E-01
9	5.167E-02	6.310E-01	3.939E+00	1.302E+00	1.291E+01	1.302E+00
10	6.502E-02	7.940E-01	5.759E+00	2.613E+00	1.407E+01	2.613E+00
11	8.189E-02	1.000E+00	9.305E+00	5.074E+00	1.336E+01	5.029E+00
12	1.032E-01	1.260E+00	1.574E+01	8.685E+00	1.850E+01	5.003E+00
13	1.294E-01	1.580E+00	2.229E+01	1.381E+01	2.231E+01	3.909E+00
14	1.638E-01	2.000E+00	2.277E+01	1.655E+01	2.315E+01	5.089E+00
15	2.055E-01	2.510E+00	2.217E+01	1.671E+01	2.282E+01	6.861E+00
16	2.588E-01	3.160E+00	1.788E+01	1.273E+01	1.888E+01	7.078E+00
17	3.259E-01	3.980E+00	1.114E+01	7.616E+00	1.430E+01	6.334E+00
18	4.102E-01	5.010E+00	6.721E+00	4.609E+00	9.469E+00	4.152E+00
19	5.167E-01	6.310E+00	5.690E+00	4.491E+00	7.375E+00	3.430E+00
20	6.502E-01	7.940E+00	4.739E+00	3.725E+00	5.151E+00	3.097E+00
21	8.189E-01	1.000E+01	3.120E+00	2.478E+00	3.905E+00	2.224E+00
22	1.032E+00	1.260E+01	2.544E+00	2.170E+00	2.694E+00	1.829E+00
23	1.294E+00	1.580E+01	1.903E+00	1.754E+00	2.055E+00	1.656E+00
24	1.638E+00	2.000E+01	1.440E+00	1.331E+00	1.503E+00	1.303E+00
25	2.055E+00	2.510E+01	1.163E+00	1.057E+00	1.206E+00	1.039E+00
26	2.588E+00	3.160E+01	8.642E-01	8.245E-01	9.416E-01	8.100E-01
27	3.259E+00	3.980E+01	6.804E-01	6.527E-01	6.994E-01	6.440E-01
28	4.102E+00	5.010E+01	5.356E-01	5.099E-01	5.432E-01	5.016E-01
29	5.167E+00	6.310E+01	4.151E-01	4.050E-01	4.277E-01	3.990E-01
30	6.502E+00	7.940E+01	3.313E-01	3.220E-01	3.352E-01	3.190E-01
31	8.189E+00	1.000E+02	2.591E-01	2.526E-01	2.642E-01	2.507E-01
32	1.032E+01	1.260E+02	2.027E-01	1.995E-01	2.063E-01	1.983E-01
33	1.294E+01	1.580E+02	1.616E-01	1.584E-01	1.624E-01	1.574E-01
34	1.639E+01	2.000E+02	1.267E-01	1.256E-01	1.276E-01	1.246E-01
35	2.055E+01	2.510E+02	1.002E-01	9.963E-02	1.011E-01	9.924E-02
36	2.588E+01	3.160E+02	7.915E-02	7.879E-02	7.995E-02	7.856E-02
37	3.259E+01	3.980E+02	6.274E-02	6.236E-02	6.334E-02	6.220E-02
38	4.102E+01	5.010E+02	4.967E-02	4.950E-02	4.991E-02	4.941E-02
39	5.167E+01	6.310E+02	3.933E-02	3.922E-02	3.945E-02	3.915E-02
40	6.502E+01	7.940E+02	3.117E-02	3.107E-02	3.128E-02	3.093E-02
41	8.189E+01	1.000E+03	2.460E-02	2.441E-02	2.477E-02	2.440E-02

Table 5. a_{32} Retrieval Results for Bimodal Distribution

A32 AND TOTAL ERROR RESULTS

FIRST ROW=VALUE LB=LOWER BRANCH
SECOND ROW=RMS ERROR UB=UPPER BRANCH
THIRD ROW=PERCENT ERROR OB=UNDETERMINED BRANCH

ZERO=NOT APPLICABLE

TRANS	L(CM)	C(G/CM3)	D(G/CM3)	Q/A(MU-1)	A32(MU) RMS ERROR BOUNDS LB	UB	OB	A32(MU) FULL ERROR BOUNDS LB	UB	OB
8.800E-01	1.000E+01	5.500E-07	3.700E+00	1.147E+01	1.03E-01	2.96E-01	2.05E-01	0.	0.	2.01E-01
0.	0.	0.	0.	0.	1.37E-02	2.52E-02	1.16E-01	0.	0.	1.64E-01
0.	0.	0.	0.	0.	1.34E+01	8.50E+00	5.67E+01	0.	0.	8.17E+01
8.800E-01	1.000E+01	5.500E-07	3.700E+00	1.147E+01	1.04E-01	3.02E-01	2.05E-01	0.	0.	2.01E-01
3.000E-02	1.000E+00	5.500E-08	4.000E-01	3.676E+00	3.32E-02	7.38E-02	1.16E-01	0.	0.	1.64E-01
2.667E+01	1.000E+01	1.000E+01	1.081E+01	3.206E+01	3.19E+01	2.44E+01	5.67E+01	0.	0.	8.17E+01



Fig. 4. $\bar{Q}/a_{32}$ Bounds - Bimodal Distribution, rms Error



Fig. 5. $\bar{Q}/a_{32}$ Bounds - Bimodal Size Distribution, Full-Range Error

APPENDIX

A32CODE Listing

```
      PROGRAM A32CODE(INPUT,OUTPUT,TAPE2,TAPE3,TAPE5=INPUT,TAPE6=OUTPUT)
C
      DIMENSION YLMAX(2),YUMAX(2)
      DIMENSION AUB(2),DUB(2),EUB(2)
      DIMENSION ALB(2),DLB(2),ELB(2)
      DIMENSION AOB(2),DOB(2),EOB(2)
      REAL N1,N2,K1,K2,NINDEX(10),KINDEX(10),L
      COMMON/QDATA/X32(41),XXXX1(8873)
      COMMON/QFUNC/QBAR(41),QSIG(41),QMIN(41),QMAX(41)
      COMMON/BOUND/A32(41),BU(2,41),BL(2,41)
C
C
C READ AND LIST INPUT DATA
      READ(5,100) N1,N2,K1,K2,WL,DISTR
      WRITE(6,200)
      WRITE(6,201) N1,N2
      WRITE(6,202) K1,K2
      WRITE(6,203) WL
      WRITE(6,204) DISTR
C READ QFILE DATA
      CALL READQ(DISTR)
C
C SET UP VARIATION ARRAYS
      NN=5
      KK=5
      DELN=(N2-N1)/(NN-1.)
      DO 1 N=1,NN
    1 NINDEX(N)=N1+(N-1.)*DELN
      DELK=(K2-K1)/(KK-1.)
      DO 2 K=1,KK
    2 KINDEX(K)=K1+(K-1.)*DELK
C SET UP A32 ARRAY AND INITIALIZE FOR UPPER AND LOWER BOUNDS SEARCH
      CC=6.283185308/WL
      DO 3 I=1,41
      A32(I)=X32(I)/CC
      DO 3 M=1,2
      BU(M,I)=0.
    3 BL(M,I)=1.E99
C PERFORM VARIATION OVER INDEX OF REFRACTION AND SET UP LOWER AND
C UPPER BOUNDS SEARCH
      DO 4 N=1,NN
      DO 4 K=1,KK
      CALL QINT(NINDEX(N),KINDEX(K))
      DO 4 I=1,41
      QMAX1=CC*(QBAR(I)+QSIG(I))
      QMIN1=CC*(QBAR(I)-QSIG(I))
      QMAX2=CC*QMAX(I)
      QMIN2=CC*QMIN(I)
      IF(QMIN1.LT.QMIN2) QMIN1=QMIN2
```

```
      IF(QMAX1.GT.BU(1,I)) BU(1,I)=QMAX1
      IF(QMIN1.LT.BL(1,I)) BL(1,I)=QMIN1
      IF(QMAX2.GT.BU(2,I)) BU(2,I)=QMAX2
      IF(QMIN2.LT.BL(2,I)) BL(2,I)=QMIN2
    4 CONTINUE
C PRINT BOUNDS RESULTS
      WRITE(6,300)
      WRITE(6,301)
      DO 5 I=1,41
    5 WRITE(6,302) I,A32(I),X32(I),BU(1,I),BL(1,I),BU(2,I),BL(2,I)
C FIT UPPER AND LOWER BOUND RESULTS WITH QUADRATIC SEGMENTS
      DELX=0.230258510
      CALL CURFIT(0,1,41,DELX,YLMAX(1))
      CALL CURFIT(1,1,41,DELX,YUMAX(1))
      CALL CURFIT(0,2,41,DELX,YLMAX(2))
      CALL CURFIT(1,2,41,DELX,YUMAX(2))
C PLOT BOUND RESULTS
      CALL PLOTB(YUMAX)
C
C DATA PROCESSING LOOP
      WRITE(6,400)
      WRITE(6,401)
      WRITE(6,402)
      WRITE(6,403)
      WRITE(6,404)
      WRITE(6,405)
C READ EXPERIMENTAL DATA AND COMPUTE EXPERIMENTAL ERROR
    6 READ(5,101) TR,DTR,L,DL,C,DC,D,DD
      IF(EOF(5)) 7,8
    7 STOP
    8 ET=-100.*DTR/(TR*ALOG(TR))
      EL=100.*DL/L
      EC=100.*DC/C
      ED=100.*DD/D
      Y=-4.E-4*D*ALOG(TR)/(3.*L*C)
      EY=SQRT(ET*ET+EL*EL+EC*EC+ED*ED)
      DY=EY*Y/100.
      YP=Y+DY
      YM=Y-DY
C LOOP OVER TWO CASES OF SIZE DISTRIBUTION ERROR -- M=1 IMPLIES RMS,
C M=2 IMPLIES FULL-RANGE
      DO 11 M=1,2
C INITIALIZE BRANCH RESULTS
      AUB(M)=0.
      ALB(M)=0.
      AOB(M)=0.
      DUB(M)=0.
      DLB(M)=0.
      DOB(M)=0.
      EUB(M)=0.
```

```
      ELB(M)=0.
      EOB(M)=0.
C UNDETERMINED(OR UNKNOWN) BRANCH ANALYSIS
      Z=Y
      IF(Z.LE.YUMAX(M)) GO TO 9
      Z=YM
      IF(Z.GT.YUMAX(M)) GO TO 10
    9 CALL ROOTS(1,M,41,Z,A1,A2)
      AOB(M)=(A2+A1)/2.
      DOB(M)=(A2-A1)/2.
      EOB(M)=100.*DOB(M)/AOB(M)
C LOWER AND UPPER BRANCH ANALYSIS
   10 IF(YP.GT.YLMAX(M)) GO TO 11
      CALL ROOTS(1,M,41,YP,A1UP,A2UP)
      CALL ROOTS(0,M,41,YP,A1LP,A2LP)
      CALL ROOTS(1,M,41,YM,A1UM,A2UM)
      CALL ROOTS(0,M,41,YM,A1LM,A2LM)
      ALB(M)=(A1UP+A1LP+A1UM+A1LM)/4.
      DLB(M)=(A1LP-A1UM)/2.
      ELB(M)=100.*DLB(M)/ALB(M)
      AUB(M)=(A2LP+A2UP+A2LM+A2UM)/4.
      DUB(M)=(A2UM-A2LP)/2.
      EUB(M)=100.*DUB(M)/AUB(M)
   11 CONTINUE
C PRINT RESULTS AND CONTINUE DATA PROCESSING LOOP
      WRITE(6,406)
      WRITE(6,407) TR,L,C,D,Y,(ALB(M),AUB(M),AOB(M),M=1,2)
      WRITE(6,407) DTR,DL,DC,DD,DY,(DLB(M),DUB(M),DOB(M),M=1,2)
      WRITE(6,407) ET,EL,EC,ED,EY,(ELB(M),EUB(M),EOB(M),M=1,2)
      GO TO 6
C
C
  100 FORMAT(5E10.0,9X,A1)
  101 FORMAT(8E10.0)
  200 FORMAT(1H1,/4X,*A32CODE RESULTS* /)
  201 FORMAT(8X,*NLIMITS*,11X,*N1=*,1PE10.3,3X,*N2=*,E10.3   )
  202 FORMAT(8X,*KLIMITS*,11X,*K1=*,1PE10.3,3X,*K2=*,E10.3 /)
  203 FORMAT(8X,*WAVELENGTH=*,F8.4,* MICRON*)
  204 FORMAT(8X,*DISTRIBUTION=*,11X,A1 ///)
  300 FORMAT(4X,*Q/A32 BOUNDS (UNIT=1/MU)* /)
  301 FORMAT(5X,*I*,4X,*A32(MU)*,7X,*X32*,7X,*AVE+SIG*,5X,*AVE-SIG*,7X,
     X *MAX*,9X,*MIN* /)
  302 FORMAT(I6,1P6E12.3)
  400 FORMAT(1H1,/3X,*A32 AND TOTAL ERROR RESULTS* /)
  401 FORMAT(6X,*FIRST ROW=VALUE                LB=LOWER BRANCH*)
  402 FORMAT(6X,*SECOND ROW=RMS ERROR           UB=UPPER BRANCH*)
  403 FORMAT(6X,*THIRD ROW=PERCENT ERROR        OB=UNDETERMINED BRANCH*)
  404 FORMAT(/6X,*ZERO=NOT APPLICABLE* /)
  405 FORMAT(5X,*TRANS*,5X,*L(CM)*,4X,*C(G/CM3)*,2X,*D(G/CM3)*,1X,
     X *Q/A(MU-1)*,2X,*A32(MU) RMS ERROR BOUNDS*,3X,*A32(MU) FULL ERROR
```

```
      XBOUNDS* /56X,*LB*,7X,*UB*,7X,*OB*,7X,*LB*,7X,*UB*,7X,*OB* /)
  406 FORMAT(/)
  407 FORMAT(2X,1P5E10.3,6E9.2)
      END
```

```
      SUBROUTINE READQ(DISTR)
C
C     READ QFILE DATA
C
      REAL NINDEX(9),KINDEX(6)
      DIMENSION X32(41)
      DIMENSION QBAR(9,6,41),QSIG(9,6,41),QMIN(9,6,41),QMAX(9,6,41)
      COMMON/QDATA/X32,NN,NK,NINDEX,KINDEX,QBAR,QSIG,QMIN,QMAX
C
      IF(DISTR.EQ.1HU) M=2
      IF(DISTR.EQ.1HB) M=3
      READ(M,100) (X32(I),I=1,41)
      READ(M,101) NN,NK
      DO 1 N=1,NN
      DO 1 K=1,NK
      READ(M,100) NINDEX(N),KINDEX(K)
      READ(M,100) (QBAR(N,K,I),I=1,41)
      READ(M,100) (QSIG(N,K,I),I=1,41)
      READ(M,100) (QMIN(N,K,I),I=1,41)
      READ(M,100) (QMAX(N,K,I),I=1,41)
    1 CONTINUE
      RETURN
C
  100 FORMAT(8E10.0)
  101 FORMAT(2I10)
      END
```

```
      SUBROUTINE QINT(NINDEX,KINDEX)
C
C         INTERPOLATE ON QFILE DATA TO GET QDATA AT NINDEX AND KINDEX
C
C
      DIMENSION FBAR(2),FSIG(2),FMIN(2),FMAX(2)
      REAL NINDEX,KINDEX,NFILE(9),KFILE(6)
      DIMENSION GBAR(9,6,41),GSIG(9,6,41),GMIN(9,6,41),GMAX(9,6,41)
      COMMON/QDATA/XXXXX(41),NN,NK,NFILE,KFILE,GBAR,GSIG,GMIN,GMAX
      COMMON/QFUNC/QBAR(41),QSIG(41),QMIN(41),QMAX(41)
C
C LOCATE FILE ENTRY POINTS
      DO 1 N=1,NN
      IF(NFILE(N).GT.NINDEX) GO TO 2
    1 CONTINUE
      N=NN
    2 DO 3 K=1,NK
      IF(KFILE(K).GT.KINDEX) GO TO 4
    3 CONTINUE
      K=NK
C SIZE PARAMETER LOOP
    4 DO 6 I=1,41
C INTERPOLATE ON N (LINEAR-LINEAR)
      DO 5 L=1,2
      M=K-2+L
      DEL=(NINDEX-NFILE(N-1))/(NFILE(N)-NFILE(N-1))
      FBAR(L)=GBAR(N-1,M,I)+(GBAR(N,M,I)-GBAR(N-1,M,I))*DEL
      FSIG(L)=GSIG(N-1,M,I)+(GSIG(N,M,I)-GSIG(N-1,M,I))*DEL
      FMIN(L)=GMIN(N-1,M,I)+(GMIN(N,M,I)-GMIN(N-1,M,I))*DEL
    5 FMAX(L)=GMAX(N-1,M,I)+(GMAX(N,M,I)-GMAX(N-1,M,I))*DEL
C INTERPOLATE ON K (LOG-LINEAR)
      DEL=(KINDEX-KFILE(K-1))/(KFILE(K)-KFILE(K-1))
      QBAR(I)=FBAR(1)*(FBAR(2)/FBAR(1))**DEL
      QSIG(I)=FSIG(1)*(FSIG(2)/FSIG(1))**DEL
      QMIN(I)=FMIN(1)*(FMIN(2)/FMIN(1))**DEL
    6 QMAX(I)=FMAX(1)*(FMAX(2)/FMAX(1))**DEL
      RETURN
      END
```

```
      SUBROUTINE CURFIT(IBOUND,M,N,DELX,YMAX)
C
C          QUADRATIC CURVE FIT TO Q/A32 VS A32 RESULTS
C
      COMMON/BOUND/XXXXX(41),BU(2,41),BL(2,41)
      COMMON/COEFF/ACU(2,40),BCU(2,40),CCU(2,40),ACL(2,40),BCL(2,40),
     X CCL(2,40)
C
C LOOP TO OBTAIN QUADRATIC COEFFICIENTS FOR ALL A INTERVALS AND
C DETERMINE ABSOLUTE MAXIMUM
      YMAX=0.
      NN=N-2
      DO 5 I=1,NN
      IF(IBOUND.EQ.1) GO TO 1
      Y1=BL(M,I)
      Y2=BL(M,I+1)
      Y3=BL(M,I+2)
      GO TO 2
    1 Y1=BU(M,I)
      Y2=BU(M,I+1)
      Y3=BU(M,I+2)
    2 A=ALOG(Y1*Y3/Y2**2)/(2.*DELX**2)
      B=ALOG(Y2**4/(Y3*Y1**3))/(2.*DELX)
      C=ALOG(Y1)
      IF(A.GE.0.) GO TO 3
      TEST=-B/(2.*A)
      IF(TEST.LT.  0.) GO TO 3
      IF(TEST.GT.DELX) GO TO 3
      YE=EXP(C-B*B/(4.*A))
      IF(YE.GT.YMAX) YMAX=YE
    3 CONTINUE
      IF(IBOUND.EQ.1) GO TO 4
      ACL(M,I)=A
      BCL(M,I)=B
      CCL(M,I)=C
      GO TO 5
    4 ACU(M,I)=A
      BCU(M,I)=B
      CCU(M,I)=C
    5 CONTINUE
      IF(IBOUND.EQ.1) GO TO 6
      ACL(M,N-1)=ACL(M,NN)
      BCL(M,N-1)=BCL(M,NN)
      CCL(M,N-1)=CCL(M,NN)
      RETURN
    6 ACU(M,N-1)=ACU(M,NN)
      BCU(M,N-1)=BCU(M,NN)
      CCU(M,N-1)=CCU(M,NN)
      RETURN
      END
```

```
      SUBROUTINE PLOTB(YUMAX)
C
C         PLOT UPPER AND LOWER Q/A32 VS A32 CURVES
C
      DIMENSION AP(41),YU1(41),YU2(41),YL1(41),YL2(41)
      DIMENSION YUMAX(2)
      COMMON/BOUND/A32(41),BU(2,41),BL(2,41)
C
C SET UP PLOT LIMITS
      NPOINTS=41
      AMIN=IFIX(ALOG10(A32(1)))-1.
      AMAX=IFIX(ALOG10(A32(41)))+1.
      Y1MAX=IFIX(ALOG10(YUMAX(1)))+1.
      Y1MIN=Y1MAX-3.
      Y2MAX=IFIX(ALOG10(YUMAX(2)))+1.
      Y2MIN=Y2MAX-3.
C SET UP PLOT VARIABLES
      DO 1 I=1,41
      AP(I)=ALOG10(A32(I))
      YU1(I)=ALOG10(BU(1,I))
      YL1(I)=ALOG10(BL(1,I))
      YU2(I)=ALOG10(BU(2,I))
      YL2(I)=ALOG10(BL(2,I))
      IF(YU1(I).LT.Y1MIN) YU1(I)=Y1MIN
      IF(YL1(I).LT.Y1MIN) YL1(I)=Y1MIN
      IF(YU2(I).LT.Y2MIN) YU2(I)=Y2MIN
      IF(YL2(I).LT.Y2MIN) YL2(I)=Y2MIN
    1 CONTINUE
C
C PLOT YU1(I) AND YL1(I) VS. AP(I)   I=1,2,...,NPOINTS  ON LINEAR-
C LINER PLOT TO GET UPPER AND LOWER BOUND CURVES APPROPRIATE TO RMS
C ERROR IN SIZE DISTRIBUTION
C
C PLOT YU2(I) AND YL2(I) VS. AP(I) TO GET CORRESPONDING RESULT FOR
C FULL RANGE ERROR IN SIZE DISTRIBUTION
C
C SUGGESTED X-AXIS TITLE ---  LOG10 A32(MICRON)
C SUGGESTED Y-AXIS TITLE ---  LOG10 Q/A32(1/MICRON)
C
C SUGGESTED TITLE FOR FIRST PLOT --- ERROR BOUNDS FOR RMS SIZE ERROR
C SUGGESTED TITLE FOR SECOND PLOT -- ERROR BOUNDS FOR FULL-RANGE SIZE
C                                    ERROR
C DELETE ABOVE COMMENTS AND INSERT YOUR PLOTTING ROUTINES HERE
      RETURN
      END
```

```
      SUBROUTINE ROOTS(IBOUND,M,N,Y,A1,A2)
C
C         DETERMINE LOWEST AND HIGHEST INTERSECTIONS OF Y WITH
C         BOUND CURVE
C
      DIMENSION AR(10)
      COMMON/BOUND/A32(41),XXXXX(164)
      COMMON/COEFF/ACU(2,40),BCU(2,40),CCU(2,40),ACL(2,40),BCL(2,40),
     X CCL(2,40)
      DATA DELX/0.230258510/
C
C SEARCH FOR ROOTS IN ALL A32 INTERVALS
      NR=0
      NN=N-1
      DO 4 I=1,NN
      IF(IBOUND.EQ.1) GO TO 1
      A=ACL(M,I)
      B=BCL(M,I)
      C=CCL(M,I)
      GO TO 2
    1 A=ACU(M,I)
      B=BCU(M,I)
      C=CCU(M,I)
    2 TEST=B*B-4.*A*(C-ALOG(Y))
      IF(TEST.LT.0.) GO TO 4
      TEST1=-(B-SQRT(TEST))/(2.*A)
      TEST2=-(B+SQRT(TEST))/(2.*A)
      IF(TEST1.LT.0.) GO TO 3
      IF(TEST1.GE.DELX) GO TO 3
      NR=NR+1
      AR(NR)=A32(I)*EXP(TEST1)
    3 IF(TEST2.LT.0.) GO TO 4
      IF(TEST2.GE.DELX) GO TO 4
      NR=NR+1
      AR(NR)=A32(I)*EXP(TEST2)
    4 CONTINUE
C FIND LOWEST AND HIGHEST ROOTS
      IF(NR.NE.0) GO TO 5
      A1=0.
      A2=0.
      RETURN
    5 IF(NR.NE.1) GO TO 6
      A1=0.
      A2=AR(1)
      RETURN
    6 A1=1.E99
      A2=0.
      DO 7 L=1,NR
      IF(AR(L).LT.A1) A1=AR(L)
```

```
      IF(AR(L).GT.A2) A2=AR(L)
    7 CONTINUE
      RETURN
      END
```

END

FILMED

10-84

DTIC